
Appendix A. User guideline for our developed FastMtQTL 

R package and SAS program (integrated SAS and R 

codes), and the existing PROC QTL SAS add-in package 

 

 

1. Example data files 

R dataset: The simulated R dataset (SimData.RData) for multi-trait QTL analysis is 

available at www.bbcba.org/FastMtQTL 

 

CSV file: The simulated dataset in CSV format (SimData.csv) is available at 

www.bbcba.org/FastMtQTL 

 

 

2. FastMtQTL R package 

1. First install the qtl (version: 1.46-2) and FastMtQTL (our developed package) R-packages 

in R-3.6.3 using the following codes. The binary (.zip) file of FastMtQTL package is 

available at www.bbcba.org/FastMtQTL 

install.packages("qtl") 

install.packages("E:/MVN-based_QTL_Mapping/pgm/FastMtQTL_0.01.zip" 

                                  , repos = NULL, type="source") 

Note that the location “E:/MVN-based_QTL_Mapping/pgm/FastMtQTL_0.01.zip” may 

need to be changed depending at which location you store the “FastMtQTL_0.01.zip” 

file.  

2. Now, load the qtl and FastMtQTL libraries in your R session using the following codes. 

library(qtl) 

library(FastMtQTL) 

3. Now, read/load the example R QTL dataset (SimData.RData) into R using the following 

codes. 

load("E:/FastMtQTL/Appendix/data files/SimData.RData") 

Note that the location of the data file may need to change depending on at which you 

store the data file. 

http://www.bbcba.org/
http://www.bbcba.org/
http://www.bbcba.org/

4. Now, run the bcMVNmtQTL(inData, phenoCols, Chrs, markerInterval=1.0) function of 

FastMtQTL R-package using the following codes. 

LODscores <- bcMVNmtQTL(inData=SimData,phenoCols=c(1:5),  

                                                  Chrs=c(1:13), markerInterval=1.0) 

plot(LODscores) 

 

That is all to calculate LOD statistic and create LOD score plot. 

 

 

3. SAS program (integrated SAS and R codes) 

1. Set the environment to run R from SAS (i.e., set up environmental variable for R to 

enable R for calling it from SAS). 

2. Make sure that the qtl and FastMtQTL R-packages are installed in R software. 

3. Run the following codes to calculate LOD statistic using the following SAS codes 

integrated with R codes. 

 

PROC IML; 

  TITLE "Statistic in R (integration with SAS)"; 

 

  SUBMIT/R; 

    #___________Begin of R code____________ 

 

    #install.packages("qtl", lib=rLibLoc) 

 

    #Load required R packages 

    library(qtl) 

    library(FastMtQTL) 

 

 

    #Load the simulated data into R 

    load("E:/FastMtQTL/Appendix/data files/SimData.RData") 

 

      

 

    #Perform the analysis to calculate LOD statistic 

    MVN_LODscores<-bcMVNmtQTL(inData=SimData,phenoCols=c(1:5),  

                              Chrs=c(1:13), markerInterval=1.0) 

 

    #Save the final count dataset permanently 

    save(MVN_LODscores, file = "E:/FastMtQTL/Appendix/data files/MVN_LODscores.RData") 

 

        #____________End of R code_____________ 

 

  ENDSUBMIT; 

QUIT; /*Use Quit command to end the IML procedure*/ 

4. PROC QTL SAS add-in package 

First install the PROC QTL package in your Windows based system. 

Data preparation 

Read the CSV data file (SimData.csv) into SAS using the following SAS codes. Note that 

you have to change the location/path of the data file according to your data file location. 

SimData.csv file contains phenotypic data and genotypic data along with the genetic map.  

 

PROC IMPORT OUT=WORK.BC_SimData  

     DATAFILE="E:\FastMtQTL\Appendix\data files\SimData.csv"  

     DBMS=CSV REPLACE; 

     GETNAMES=YES; 

     DATAROW=2; 

RUN; 

 

Now, create a SAS dataset (Bc_simdata_map_final) of the genetic map using the 

BC_SimData SAS dataset using the following SAS codes. 

 

DATA Bc_simdata_map; 

  SET Bc_simdata(DROP=Pheno1--Pheno5); 

  IF _N_ > 2 THEN DELETE; 

RUN; 

 

PROC TRANSPOSE DATA=Bc_simdata_map OUT=Bc_simdata_map_trns; 

  VAR _ALL_; 

RUN; 

 

DATA Bc_simdata_map_final(DROP=_NAME_ COL1 COL2); 

  SET Bc_simdata_map_trns; 

 

  marker=_NAME_; 

  position = INPUT(COL2,BEST12.); 

  chromosom = INPUT(COL1,BEST12.); 

RUN; 

 

Now, create dataset (Bc_simdata_Genopheno) that contains the phenotypic and genotypic 

information using the following SAS codes. 

 

DATA Bc_simdata_Genopheno; 

  SET Bc_simdata; 

  IF _N_ IN(1,2) THEN DELETE; 

RUN; 

Maximum likelihood (ML) based multi-trait QTL analysis 

The followings are the SAS codes for genome-wide QTL searching at each 1 cM 

(STEP=1.0) flanking marker interval using ML method (METHOD='ml') of multi-trait 

QTL analysis. 

PROC QTL DATA=Bc_simdata_Genopheno  

         MAP=Bc_simdata_map_final  

         OUT=result_ML 

         METHOD='ml' STEP=1.0; 

  MODEL Pheno1 Pheno2 Pheno3 Pheno4 Pheno5=; 

  MATINGTYPE 'BC'; 

  GENOTYPE A1A1='A' A1A2='H'; 

  ESTIMATE 'additive'=1 0; 

RUN; 

QUIT;  

 

Now calculate log of odds (LOD) score as follows. 

DATA result_ML_LOD; 

  SET result_ML; 

 

  index=_N_; 

  LOD = 0.2171*LRT; 

RUN; 

 

Least squares (LS) based multi-trait QTL analysis 

The followings are the SAS codes for genome-wide QTL searching at each 1 cM 

(STEP=1.0) flanking marker interval using LS method (METHOD='ls') of multi-trait 

QTL analysis. 

PROC QTL DATA=Bc_simdata_Genopheno  

         MAP=Bc_simdata_map_final  

         OUT=result_Reg 

         METHOD='ls' STEP=1.0; 

  MODEL Pheno1 Pheno2 Pheno3 Pheno4 Pheno5=; 

  MATINGTYPE 'BC'; 

  GENOTYPE A1A1='A' A1A2='H'; 

  ESTIMATE 'additive'=1 0; 

RUN; 

QUIT; 

 

Now calculate log of odds (LOD) score as follows. 

DATA result_reg_LOD; 

  SET result_Reg; 

 

  index=_N_; 

  LOD = 0.2171*LRT; 

RUN; 


